

Pensieve Header: A short program to compute cycle decompositions and immanants.

```

p = Permutations [{1, 2, 3, 4, 5}] [[69]]
{3, 5, 2, 1, 4}

c = Range[5]
{1, 2, 3, 4, 5}

q = p
{3, 5, 2, 1, 4}

c = MapThread[Min, {c, c[[q]]}]
{1, 2, 2, 1, 4}

q = q[[q]]
{2, 4, 5, 3, 1}

c = MapThread[Min, {c, c[[q]]}]
{1, 1, 2, 1, 1}

q = q[[q]]
{4, 3, 1, 5, 2}

c = MapThread[Min, {c, c[[q]]}]
{1, 1, 1, 1, 1}

Cycles[p_List] := Cycles[p] = Module[
  {q = p, c},
  FixedPoint[
    (c = MapThread[Min, {#, #[[q]]}]; q = q[[q]]; c) &,
    Range[Length[p]]
  ];
  Reverse[Sort[Last /@ Tally[c]]]
]

Tally[Cycles /@ Permutations[Range[6]]]
{{{1, 1, 1, 1, 1, 1}, 1}, {{2, 1, 1, 1, 1}, 15},
 {{3, 1, 1, 1}, 40}, {{2, 2, 1, 1}, 45}, {{4, 1, 1}, 90}, {{3, 2, 1}, 120},
 {{5, 1}, 144}, {{2, 2, 2}, 15}, {{4, 2}, 90}, {{3, 3}, 40}, {{6}, 120}}
```

```

Immanant[M_?MatrixQ] := Module[
  {n = Length[M], ps, p},
  ps = Permutations[Range[n]];
  Sum[
    p = ps[[i]];
    (YT@@Cycles[p]) * (Times @@ (M[[#, p[[#]]]] & /@ p)),
    {i, n!}
  ] /; Dimensions[M] == {n, n}
]

```

```
Immanant[{{1, 2}, {3, 4}}]
```

```
6 YT[2] + 4 YT[1, 1]
```

```
Immanant[Partition[Range[9], 3]]
```

```
180 YT[3] + 225 YT[2, 1] + 45 YT[1, 1, 1]
```